11 連立方程式と行列演算

物理学をはじめ、多くの分野で連立方程式を解くことが必要となる。本章では、多くの自由度を持つ線形現象を調べる際に必要となる n 元連立 1 次方程式の解法について述べる。具体的には、定数 $a_{ij}(i,j=1,\cdots,n)$ と $b_i(i=1,\cdots,n)$ が与えられた時、次の n 個の式を同時に満足する $x_i(i=1,\cdots,n)$ を求める。

$$\begin{aligned} a_{11}x_1+a_{12}x_2+\cdots+a_{1n}x_n&=b_1\\ a_{21}x_1+a_{22}x_2+\cdots+a_{2n}x_n&=b_2\\ &\vdots\\ a_{n1}x_1+a_{n2}x_2+\cdots+a_{nn}x_n&=b_n \end{aligned} \tag{11.1}$$

連立方程式は、係数行列 A 及びベクトル $\boldsymbol{b},\boldsymbol{x}$ を、

$$A=\begin{pmatrix} a_{11} & \ldots & a_{1n}\\ \vdots & \ddots & \vdots\\ a_{n1} & \ldots & a_{nn}\end{pmatrix},\quad \boldsymbol{b}=\begin{pmatrix} b_1\\ \vdots\\ b_n\end{pmatrix},\quad \boldsymbol{x}=\begin{pmatrix} x_1\\ \vdots\\ x_n\end{pmatrix} \tag{11.2}$$

とおくと、

$$A\boldsymbol{x}=\boldsymbol{b} \tag{11.3}$$

と表すことができる。従って、連立方程式の解を求めるということは、

$$\boldsymbol{x}=A^{-1}\boldsymbol{b} \tag{11.4}$$

であるので、行列 A の逆行列 A^{-1} を求める事でもある。

11.1 消去法

まず、我々が筆算で連立方程式を解くときに行っている事を整理する。具体的な例として、

$$2x-1y-1z = +1 \tag{11.5}$$
$$3x-2y+2z = -3 \tag{11.6}$$
$$1x-2y+1z = -4 \tag{11.7}$$

を考える。まず、式 (11.5) を用いて、式 (11.6)、(11.7) の x の係数を 0 にする。

$$2x-1y-1z = +1 \quad \ldots(11.5) \tag{11.8}$$
$$0x-\frac{1}{2}y+\frac{7}{2}z = -\frac{9}{2} \quad \ldots(11.6)-(11.5)\times\frac{3}{2} \tag{11.9}$$
$$0x-\frac{3}{2}y+\frac{3}{2}z = -\frac{9}{2} \quad \ldots(11.7)-(11.5)\times\frac{1}{2} \tag{11.10}$$

次に、式 (11.9) を用いて式 (11.10) の y の係数を 0 にする。

$$2x-1y-1z = +1 \quad \ldots(11.5) \tag{11.11}$$
$$0x-\frac{1}{2}y+\frac{7}{2}z = -\frac{9}{2} \quad \ldots(11.9) \tag{11.12}$$
$$0x+0y-9z = +9 \quad \ldots(11.10)-(11.9)\times 3 \tag{11.13}$$

この後、式 (11.13)→ 式 (11.12) → 式 (11.11) とさかのぼることにより、順次 $z = -1$、$y = 2$、$x = 1$ を得る。

この例を一般化したものが、次に述べる Gauss の消去法である。

11.2 Gauss の消去法

n 元連立 1 次方程式 $A\boldsymbol{x} = \boldsymbol{b}$ の解を Gauss の消去法で求める手順は以下の通り。

1. 係数行列 A と列ベクトル $\boldsymbol{b}$ からなる $n \times (n+1)$ の行列 $\tilde{A}$ を作る。

$$\tilde{A} = \begin{pmatrix} a_{11} & a_{12} & a_{13} & \dots & a_{1n} & b_1 \\ a_{21} & a_{22} & a_{23} & \dots & a_{2n} & b_2 \\ a_{31} & a_{32} & a_{33} & \dots & a_{2n} & b_2 \\ \vdots & \vdots & \ddots & \ddots & \vdots & \vdots \\ a_{n1} & a_{n2} & a_{n3} & \dots & a_{nn} & b_n \end{pmatrix} \tag{11.14}$$

2. $\tilde{A}$ を変換し、係数行例の左下半分の非対角要素が 0 であるような行列 $\tilde{A}'$

$$\tilde{A}' = \begin{pmatrix} a'_{11} & a'_{12} & a'_{13} & \dots & a'_{1n} & b'_1 \\ 0 & a'_{22} & a'_{23} & \dots & a'_{2n} & b'_2 \\ 0 & 0 & a'_{33} & \dots & a'_{3n} & b'_3 \\ \vdots & \vdots & \ddots & \ddots & \vdots & \vdots \\ 0 & 0 & \dots & 0 & a'_{nn} & b'_n \end{pmatrix} \tag{11.15}$$

に変換する。この過程を前段過程という。

この前段過程において、

$$c = \frac{a_{ji}}{a_{ii}} \quad (j > i) \tag{11.16}$$

として c の値を決め、

$$a''_{jk} = a_{jk} - c * a_{ik} \quad (k = i+1, \cdots, n) \tag{11.17}$$

とする変換、すなわち a_{ji} を 0 にする変換を、

除数 a_{ii} をピボット (pivot、軸要素) にした掃き出し

という。

3. 変換後、まず、最下行 (n 行) が与える式

$$a'_{nn}x_n = b'_n \tag{11.18}$$

から x_n の値を求める ($x_n = b'_n/a'_{nn}$)。

次に、第 $(n-1)$ 行が与える式

$$a'_{n-1,n-1}x_{n-1} + a'_{n-1,n}x_n = b'_{n-1} \tag{11.19}$$

に既に求められた x_n を代入して、x_{n-1} を求める。

このような手順を繰り返すと、一般に x_i の値は、すでに求められた $x_{i+1},\ldots,x_n$ を用いて、

$$x_i = \frac{b'_i - \sum_{j=i+1}^{n} a'_{ij}x_j}{a'_{ii}} \tag{11.20}$$

で与えられる。この過程を後段過程という。

Gauss の消去法のサンプルプログラム

Gauss の消去法のサンプルプログラムは以下の通り。プログラムは、

/home/teacher/z6wt01in/SAMPLE/gauss.f

として置いてある。

```
      subroutine gauss(n,a,b)
      implicit none
c******************************************
c Gauss の消去法により n 元連立方程式
c    Ax = b
c の解を求める。
c 入力
c    n: 次元
c    a: 係数行列 A の要素 (n x n)
c    b: ベクトル b の要素 (n)
c 出力
c    a: 行列 A を上三角行列に変換したもの
c       (左下の非対角要素は意味が無い)
c    b: 解 x
c******************************************
c input/output:
      integer n
      real    a(n,n),b(n)
c local
      integer i,j,k
c begin:
      do i=1,n
         do j=i+1,n
            a(j,i)=a(j,i)/a(i,i)   ! Pivot:c=a_ji/a_ii
            do k=i+1,n
               a(j,k)=a(j,k)-a(j,i)*a(i,k)
            end do
               b(j)=b(j)-a(j,i)*b(i)
         end do
      end do
```

```
do i=n,1,-1
   do j=i+1,n
      b(i)=b(i)-a(i,j)*b(j)
   end do
   b(i)=b(i)/a(i,i)          ! Ax=b の解 x
end do
return
end
```

練習

Gauss の消去法により、筆算で解いた 3 元連立方程式を数値計算により解け。より高い次元の連立方程式も解いてみよ。

11.3 Gauss-Jordan の消去法

Gauss の消去法は、前段過程において行列 $\tilde{A}$ を、左下の非対角成分が 0 の行列 $\tilde{A}'$ に変換した。つまり、a_{ii} をピボットに選択し、下方の $j = i+1, \ldots, n$ 行の要素を対象に掃き出しを行った (0 にした)。

この際に同時に上方の $j = 1, \ldots, i-1$ 行の要素に対しても同様な掃き出しを行い、対角要素 $a_{ii}(i = 1, \ldots, n)$ のみを残す変換が Gauss-Jordan の消去法と呼ばれるものである。この際、対角要素が全て 1 となるように規格化する操作も同時に行う。

この手順を筆算で求めた 3 元連立方程式に対して具体的に説明すると以下のようになる。

1. まず、式 (11.5) を定数倍して x の係数 (a_{11} に相当) を 1 にする。この式 (11.21) を用いて、式 (11.6)、(11.7) の x の係数を 0 にする。

$$1x - \frac{1}{2}y - \frac{1}{2}z = +\frac{1}{2} \quad \ldots(11.5) \times \frac{1}{2} \tag{11.21}$$

$$0x - \frac{1}{2}y + \frac{7}{2}z = -\frac{9}{2} \quad \ldots(11.6) - (11.21) \times 3 \tag{11.22}$$

$$0x - \frac{3}{2}y + \frac{3}{2}z = -\frac{9}{2} \quad \ldots(11.7) - (11.21) \times 1 \tag{11.23}$$

2. 次に、式 (11.22) を定数倍して y の係数 (a_{22} に相当) を 1 にする。この式 (11.25) を用いて、式 (11.21)、(11.23) の y の係数を 0 にする。

$$1x + 0y - 4z = +5 \quad \ldots(11.21) - (11.25) \times \left(-\frac{1}{2}\right) \tag{11.24}$$

$$0x + 1y - 7z = +9 \quad \ldots(11.22) \times (-2) \tag{11.25}$$

$$0x + 0y - 9z = +9 \quad \ldots(11.23) - (11.25) \times \left(-\frac{3}{2}\right) \tag{11.26}$$

3. 最後に、式 (11.26) を定数倍して z の係数 (a_{33} に相当) を 1 にする。この式 (11.29) を用い

て、式 (11.24)、(11.25) の z の係数を 0 にする。

$$
\begin{aligned}
1x + 0y + 0z &= +1 \quad \ldots(11.24) + (11.29) \times 4 && (11.27)\\
0x + 1y + 0z &= +2 \quad \ldots(11.25) + (11.29) \times 7 && (11.28)\\
0x + 0y + 1z &= -1 \quad \ldots(11.26) \times \left(-\frac{1}{9}\right) && (11.29)
\end{aligned}
$$

結果から分かるように、この方法では求める解 $\boldsymbol{x}$ が自動的に得られている。

Gauss-Jordan の消去法のサンプルプログラム

Gauss-Jordan の消去法のサンプルプログラムは以下の通り。プログラムは、

`/home/teacher/z6wt01in/SAMPLE/jordan.f`

として置いてある。なお、このプログラムは解 x を求める事に特化しており、対角要素のみが 1(その他は 0) 行列 $\tilde{A}'$ を求める計算は完全には行っていない事に注意せよ (計算量を減らして、高速化を図っている)。

```
      subroutine jordan(n,f,b)
      implicit none
c******************************************
c Gauss-Jordan の消去法により n 元連立方程式
c    Ax = b
c の解を求める。
c 入力
c    n: 次元
c    f: 係数行列 A の要素 (n x n)
c    b: ベクトル b の要素 (n)
c 出力
c    b: 解 x
c******************************************
c input/output:
      integer n
      real    f(n,n),b(n)
c local:
      real    a(n,n+1)
      integer i,j,k
c begin:
c Initialize:
      do i=1,n
         do j=1,n
            a(i,j)=f(i,j)
         end do
```

```
         a(i,n+1)=b(i)
      end do
c Gauss-Jordan:
      do k=1,n
         do j=k+1,n+1
            if (j.ne.k) a(k,j)=a(k,j)/a(k,k)
         end do
         do i=1,n
            do j=k+1,n+1
               if (i.ne.k.and.j.ne.k) then
                  a(i,j)=a(i,j)-a(i,k)*a(k,j)
               endif
            end do
         end do
      end do
      do i=1,n
         b(i)=a(i,n+1)
      end do
      return
      end
```

練習

Gauss-Jordan の消去法により、筆算で解いた 3 元連立方程式を数値計算により解け。より高い次元の連立方程式も解いてみよ。

11.4 逆行列

Gauss-Jordan の消去法を利用した逆行列の計算を考える。

Gauss-Jordan の消去法は、係数行例 A を単位行列 E に変換する操作である。従って、これと同じ操作を単位行列 E に対して行えば、単位行列 E が行列 A の逆行列 A^{-1} に変換されるはずである。

練習

Gauss-Jordan の消去法を用いて、行列 A の逆行列 A^{-1} を求めるプログラムを作成せよ。

逆行列を求めるサンプルプログラム

Gauss-Jordan の消去法を用いて逆行列を求めるサンプルプログラムは以下の通り。プログラムは、

/home/teacher/z6wt01in/SAMPLE/matinv_jordan.f

として置いてある。なお、計算精度を考慮して、実数型の変数は `real*8` で宣言してあることに注意せよ。

```
      subroutine matinv_jordan(n,a)
      implicit none
c*******************************************
c Gauss-Jordan の消去法により n 次正方行列 A の
c 逆行列 A^-1 を求める。
c 入力
c    n: 次元
c    a: 行列 A の要素 (n x n)
c 出力
c    a: 逆行列 A^-1 の要素
c*******************************************
c input/output:
      integer n
      real*8  a(n,n)
      integer i,j,k
c local:
      real*8  c       ! Pivot
c begin:
      do k=1,n
         c=1.0/a(k,k)
         a(k,k)=c
         do j=1,n
            if (j.ne.k) a(k,j)=c*a(k,j)
         end do
         do i=1,n
            do j=1,n
               if (i.ne.k.and.j.ne.k) then
                  a(i,j)=a(i,j)-a(i,k)*a(k,j)
               endif
            end do
         end do
         do i=1,n
            if (i.ne.k) a(i,k)=-c*a(i,k)
         end do
      end do
      return
      end
```

練習

適当な n 次正方行列 A に対して、Gauss-Jordan の消去法を用いて、行列 A の逆行列 A^{-1} を求めよ。$AA^{-1} = A^{-1}A = E$ を満たしている事を、数値計算により確かめよ。

$\boxed{\text{課題}}$
Gauss-Jordan の消去法を用いれば

- $A\boldsymbol{x} = \boldsymbol{b}$ を満たす $\boldsymbol{x}$ を求める (連立方程式を解く)
- $AA^{-1}{=}A^{-1}A = E$ を満たす A^{-1} を求める (逆行列を求める)

という数値計算ができる。例題では上記の計算を個々に行ったが、新しい n 行 $2n+1$ 列の行列 (A が n 次正方行列とする)

$$[A|E|\boldsymbol{b}] = \begin{pmatrix} a_{11} & a_{12} & \dots & a_{1n} & 1 & 0 & \dots & 0 & b_1 \\ a_{21} & a_{22} & \dots & a_{2n} & 0 & 1 & \dots & 0 & b_2 \\ \vdots & \vdots & \ddots & \vdots & \vdots & \vdots & \ddots & \vdots & \vdots \\ a_{n1} & a_{n2} & \dots & a_{nn} & 0 & 0 & \dots & 1 & b_n \end{pmatrix} \tag{11.30}$$

を用意し、この行列に対して Gauss-Jordan 法で行った掃き出し計算を行うと、

$$[A|E|\boldsymbol{b}] \xrightarrow{\text{掃き出し計算}} [E|A^{-1}|\boldsymbol{x}] \tag{11.31}$$

となり、逆行列 A^{-1} と連立方程式の解 $\boldsymbol{x}$ を同時に得る事ができる。この処理を行うプログラムを作成し、適当な例題について正しく動作している事を確認せよ。

$\boxed{\text{課題}}$
サンプルプログラムは単純すぎて、実用上は色々と問題がある可能性がある。例えば、行列 A の対角要素 a_{ii} が 0 だと、ピボット (pivot) の計算が 0 での除算になり計算できない。また、a_{ii} が 0 でなくとも、絶対値が極端に小さいと数値計算の誤差の影響が無視できない。さらに、A の行列式が 0 になる場合の判別も行われていない。これらの問題を解決したプログラムを作成し、適当な例題について正しく動作している事を確認せよ。